

NAG Library Routine Document

F08FNF (ZHEEV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08FNF (ZHEEV) computes all the eigenvalues and, optionally, all the eigenvectors of a complex n by n Hermitian matrix A .

2 Specification

```
SUBROUTINE F08FNF(JOBZ, UPLO, N, A, LDA, W, WORK, LWORK, RWORK, INFO)
INTEGER          N, LDA, LWORK, INFO
double precision W(*), RWORK(*)
complex*16      A(LDA,*), WORK(max(1,LWORK))
CHARACTER*1      JOBZ, UPLO
```

The routine may be called by its LAPACK name *zheev*.

3 Description

The Hermitian matrix A is first reduced to real tridiagonal form, using unitary similarity transformations, and then the QR algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally) the eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: JOBZ – CHARACTER*1 *Input*
On entry: if JOBZ = 'N', compute eigenvalues only.
 If JOBZ = 'V', compute eigenvalues and eigenvectors.
Constraint: JOBZ = 'N' or 'V'.
- 2: UPLO – CHARACTER*1 *Input*
On entry: if UPLO = 'U', the upper triangular part of A is stored.
 If UPLO = 'L', the lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.
- 3: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

- 4: A(LDA,*) – **complex*16** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n Hermitian matrix A .
 If UPLO = 'U', the upper triangular part of A must be stored and the elements of the array below the diagonal are not referenced.
 If UPLO = 'L', the lower triangular part of A must be stored and the elements of the array above the diagonal are not referenced.
On exit: if JOBZ = 'V', then if INFO = 0, A contains the orthonormal eigenvectors of the matrix A .
 If JOBZ = 'N', then on exit the lower triangle (if UPLO = 'L') or the upper triangle (if UPLO = 'U') of A, including the diagonal, is destroyed.
- 5: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08FNF (ZHEEV) is called.
Constraint: $LDA \geq \max(1, N)$.
- 6: W(*) – **double precision** array *Output*
Note: the dimension of the array W must be at least $\max(1, N)$.
On exit: if INFO = 0, the eigenvalues in ascending order.
- 7: WORK(max(1, LWORK)) – **complex*16** array *Workspace*
On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 8: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08FNF (ZHEEV) is called.
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
Suggested value: for optimal performance, $LWORK \geq (nb + 1) \times N$, where nb is the optimal **block size** for F08FSF (ZHETRD).
Constraint: $LWORK \geq \max(1, 2 \times N)$.
- 9: RWORK(*) – **double precision** array *Workspace*
Note: the dimension of the array RWORK must be at least $\max(1, 3 \times N - 2)$.
- 10: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , the algorithm failed to converge; i elements of an intermediate tridiagonal form did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Further Comments

Each eigenvector is normalized so that the element of largest absolute value is real and positive.

The total number of floating-point operations is proportional to n^3 .

The real analogue of this routine is F08FAF (DSYEV).

9 Example

This example finds all the eigenvalues and eigenvectors of the Hermitian matrix

$$A = \begin{pmatrix} 1 & 2-i & 3-i & 4-i \\ 2+i & 2 & 3-2i & 4-2i \\ 3+i & 3+2i & 3 & 4-3i \\ 4+i & 4+2i & 4+3i & 4 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues and eigenvectors.

9.1 Program Text

```
*      F08FNF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NB, NMAX
      PARAMETER        (NB=64,NMAX=10)
      INTEGER          LDA, LWORK
      PARAMETER        (LDA=NMAX,LWORK=(NB+1)*NMAX)
*      .. Local Scalars ..
      DOUBLE PRECISION EERRBD, EPS
      INTEGER          I, IFAIL, INFO, J, LWKOPT, N
*      .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), WORK(LWORK)
      DOUBLE PRECISION RCONDZ(NMAX), RWORK(3*NMAX-2), W(NMAX),
+                     ZERRBD(NMAX)
*      .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL         X02AJF
*      .. External Subroutines ..
      EXTERNAL         DDISNA, X04DAF, ZHEEV
*      .. Intrinsic Functions ..
      INTRINSIC        ABS, INT, MAX
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08FNF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN

*
*          Read the upper triangular part of the matrix A from data file
*

```

```

      READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
*
*      Solve the Hermitian eigenvalue problem
*
      CALL ZHEEV('Vectors','Upper',N,A,LDA,W,WORK,LWORK,RWORK,INFO)
      LWKOPT = INT(WORK(1))
*
      IF (INFO.EQ.0) THEN
*
*          Print solution
*
          WRITE (NOUT,*) 'Eigenvalues'
          WRITE (NOUT,99999) (W(J),J=1,N)
*
          WRITE (NOUT,*)
          IFAIL = 0
          CALL X04DAF('General',' ',N,N,A,LDA,'Eigenvectors',IFAIL)
*
*          Get the machine precision, EPS and compute the approximate
*          error bound for the computed eigenvalues. Note that for
*          the 2-norm, max( abs(W(i)) ) = norm(A), and since the
*          eigenvalues are returned in descending order
*          max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)) )
*
          EPS = X02AJF()
          EERRBD = EPS*MAX(ABS(W(1)),ABS(W(N)))
*
*          Call DDISNA (F08FLF) to estimate reciprocal condition
*          numbers for the eigenvectors
*
          CALL DDISNA('Eigenvectors',N,N,W,RCONDZ,INFO)
*
*          Compute the error estimates for the eigenvectors
*
          DO 20 I = 1, N
              ZERRBD(I) = EERRBD/RCONDZ(I)
20      CONTINUE
*
*          Print the approximate error bounds for the eigenvalues
*          and vectors
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Error estimate for the eigenvalues'
          WRITE (NOUT,99998) EERRBD
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Error estimates for the eigenvectors'
          WRITE (NOUT,99998) (ZERRBD(I),I=1,N)
          ELSE
              WRITE (NOUT,99997) 'Failure in ZHEEV. INFO =', INFO
          END IF
*
*          Print workspace information
*
          IF (LWORK.LT.LWKOPT) THEN
              WRITE (NOUT,*)
              WRITE (NOUT,99996) 'Optimum complex workspace required = ',
+              LWKOPT, 'Workspace provided in WORK          = ', LWORK
          END IF
          ELSE
              WRITE (NOUT,*) 'NMAX too small'
          END IF
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
99996 FORMAT (1X,A,I5,/1X,A,I5)
      END

```

9.2 Program Data

F08FNF Example Program Data

```

4                                     :Value of N

(1.0, 0.0) (2.0, -1.0) (3.0, -1.0) (4.0, -1.0)
          (2.0, 0.0) (3.0, -2.0) (4.0, -2.0)
                    (3.0, 0.0) (4.0, -3.0)
                                (4.0, 0.0) :End of matrix A

```

9.3 Program Results

F08FNF Example Program Results

Eigenvalues

```
-4.2443 -0.6886  1.1412 13.7916
```

Eigenvectors

	1	2	3	4
1	-0.3839	0.6470	0.0179	0.3309
	-0.2941	0.0000	-0.4453	-0.1986
2	-0.4512	-0.4984	0.5706	0.3728
	0.1102	-0.1130	0.0000	-0.2419
3	0.0263	0.2949	-0.1530	0.4870
	0.4857	0.3165	0.5273	-0.1938
4	0.5602	-0.2241	-0.2118	0.6155
	0.0000	-0.2878	-0.3598	0.0000

Error estimate for the eigenvalues

```
1.5E-15
```

Error estimates for the eigenvectors

```
4.3E-16  8.4E-16  8.4E-16  1.2E-16
```
